Problem:
Score: 600 points

Problem Statement

We have N non-negative integers: A_1, A_2, ..., A_N.

Consider painting at least one and at most N-1 integers among them in red, and painting the rest in blue.

Let the beauty of the painting be the \mbox{XOR} of the integers painted in red, plus the \mbox{XOR} of the integers painted in blue.

Find the maximum possible beauty of the painting.

What is \mbox{XOR}?

The bitwise \mbox{XOR} x_1 \oplus x_2 \oplus \ldots \oplus x_n of n non-negative integers x_1, x_2, \ldots, x_n is defined as follows:

When x_1 \oplus x_2 \oplus \ldots \oplus x_n is written in base two, the digit in the 2^k's place (k \geq 0) is 1 if the number of integers among x_1, x_2, \ldots, x_n whose binary representations have 1 in the 2^k's place is odd, and 0 if that count is even.

For example, 3 \oplus 5 = 6.

Constraints

All values in input are integers.

2 \leq N \leq 10^5

0 \leq A_i < 2^{60}\ (1 \leq i \leq N)

Input

Input is given from Standard Input in the following format:

N
A_1 A_2 ... A_N

Output

Print the maximum possible beauty of the painting.

Sample Input 1

3
3 6 5

Sample Output 1

12

If we paint 3, 6, 5 in blue, red, blue, respectively, the beauty will be (6) + (3 \oplus 5) = 12.

There is no way to paint the integers resulting in greater beauty than 12, so the answer is 12.

Sample Input 2

4
23 36 66 65

Sample Output 2

188

Sample Input 3

20
1008288677408720767 539403903321871999 1044301017184589821 215886900497862655 504277496111605629 972104334925272829 792625803473366909 972333547668684797 467386965442856573 755861732751878143 1151846447448561405 467257771752201853 683930041385277311 432010719984459389 319104378117934975 611451291444233983 647509226592964607 251832107792119421 827811265410084479 864032478037725181

Sample Output 3

2012721721873704572

A_i and the answer may not fit into a 32-bit integer type.

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program xor_sum_3
  implicit none
  integer :: n, i, j, k, x(100000,60) = 0
  integer(8) :: a(100000) = 0_8, ans1 = 0_8, ans2 = 0_8, tmp
  read(*,*) n
  read(*,*) a(1:n)
  do i = 1, n
    do j = 1, 60
      if (btest(a(i),j-1)) x(i,j) = 1
    end do
  end do
  do j = 1, 60
    if (mod(sum(x(1:n,j)),2) == 1) then
      ans1 = or(ans1,lshift(1_8,j-1))
      x(1:n,j) = 0
    end if
  end do
  x(1:n,1:60) = x(1:n,60:1:-1)
  call normal_form(x(1:n,:))
  x(1:n,1:60) = x(1:n,60:1:-1)
  j = 60
  do i = 1, n
    do while (j > 0 .and. x(i,j) == 0)
      j = j-1
    end do
    if (j == 0) exit
    tmp = 0_8
    do k = 1, 60
      if (x(i,k) == 1) tmp = ibset(tmp,k-1)
    end do
    ans2 = xor(ans2,tmp)
  end do
  write(*,'(i0)') ans1+2_8*ans2
contains
  subroutine normal_form(a)
    integer, intent(inout) :: a(:,:)
    integer :: n, m, i, j, k, l
    n = size(a,1)
    m = size(a,2)
    i = 1
    do j = 1, m
      if (sum(a(i:n,j)) == 0) cycle
      if (a(i,j) == 0) then
        do k = i+1, n
          if (a(k,j) == 1) then
            do l = j, m
              a(i,l) = xor(a(i,l),a(k,l))
            end do
            exit
          end if
        end do
      end if
      do k = 1, n
        if (k == i) cycle
        if (a(k,j) == 0) cycle
        do l = j, m
          a(k,l) = xor(a(k,l),a(i,l))
        end do
      end do
      i = i+1
      if (i > n) exit
    end do
  end
end program xor_sum_3